Привет!
Есть 1С ЕРП.
Есть дополнительный реквизит с типом значения "Дата".
Нужно получить запросом только те элементы справочника, которые имеют пустую дату (ЗначениеПВХ в параметре пустой) и статус "Действует":

```1c
    ВЫБРАТЬ
    ДоговорыКонтрагентов.Ссылка КАК СсылкаНАДоговор,
    ЕСТЬNULL(ДоговорыКонтрагентовДополнительныеРеквизиты.Значение, ЗНАЧЕНИЕ(Справочник.ЗначенияСвойствОбъектов.ПустаяСсылка)) КАК ДатаСозданияДоговора
    ИЗ
    Справочник.ДоговорыКонтрагентов КАК ДоговорыКонтрагентов
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ Справочник.ДоговорыКонтрагентов.ДополнительныеРеквизиты КАК ДоговорыКонтрагентовДополнительныеРеквизиты
    ПО (ДоговорыКонтрагентов.Ссылка = ДоговорыКонтрагентовДополнительныеРеквизиты.Ссылка)
    И (ДоговорыКонтрагентовДополнительныеРеквизиты.Свойство.Имя = "ДатаСозданияДоговора_123")
    И (ДоговорыКонтрагентовДополнительныеРеквизиты.Значение = &ЗначениеПВХ)
    ГДЕ
    НЕ ДоговорыКонтрагентов.ПометкаУдаления
    И ДоговорыКонтрагентов.Статус = &Статус
    И ТИПЗНАЧЕНИЯ(ДоговорыКонтрагентовДополнительныеРеквизиты.Значение) = ТИП(ДАТА)
```

Результат запроса пустой.
Что не так делаю?
---
```1c
    ВЫБРАТЬ
    ДоговорыКонтрагентов.Ссылка КАК СсылкаНАДоговор,
    ЕСТЬNULL(ДоговорыКонтрагентовДополнительныеРеквизиты.Значение, ДАТАВРЕМЯ(1,1,1)) КАК ДатаСозданияДоговора
    ИЗ
    Справочник.ДоговорыКонтрагентов КАК ДоговорыКонтрагентов
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ДоговорыКонтрагентов.ДополнительныеРеквизиты КАК ДоговорыКонтрагентовДополнительныеРеквизиты
    ПО (ДоговорыКонтрагентов.Ссылка = ДоговорыКонтрагентовДополнительныеРеквизиты.Ссылка)
    И (ДоговорыКонтрагентовДополнительныеРеквизиты.Свойство.Имя = "ДатаСозданияДоговора_123")
    ГДЕ
    НЕ ДоговорыКонтрагентов.ПометкаУдаления
    И ДоговорыКонтрагентов.Статус = &Статус
    И ЕСТЬNULL(ДоговорыКонтрагентовДополнительныеРеквизиты.Значение, ДАТАВРЕМЯ(1,1,1)) = ДАТАВРЕМЯ(1,1,1)
```
---
Если справочнику не устанавливали доп.реквизит, то в табличной части его и не будет.
А при внутреннем соединении в результате будут только записи, которые одновременно есть в обоих таблицах.

То есть надо получить те договоры, в табличной части которых есть реквизит с пустым значением и добавить те договоры, в табличной части которых такого реквизита нет вообще.
---
в справочнике доп.реквизит виден, но не заполнен. Так как хочу написать обработку, для автозаполнения, а для этого отобрать все с пустой датой.
---
На форме виден - это не значит, что он есть в табличной части.
Когда для справочника добавляется новый доп.реквизит не выполняется перезапись всех элементов данного справочника, чтобы в табличную часть его записать доп.реквизит и пустое значение.
---
Мне кажется, все дело в параметре. А чему равен параметр ЗначениеПВХ у Вас?
---
Тип - "Дата", значение - пусть
---
```1c
    ВЫБРАТЬ
    ДоговорыКонтрагентов.Ссылка КАК СсылкаНАДоговор,
    ЕСТЬNULL(ДоговорыКонтрагентовДополнительныеРеквизиты.Значение, ДАТАВРЕМЯ(1,1,1)) КАК ДатаСозданияДоговора
    ИЗ
    Справочник.ДоговорыКонтрагентов КАК ДоговорыКонтрагентов
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ДоговорыКонтрагентов.ДополнительныеРеквизиты КАК ДоговорыКонтрагентовДополнительныеРеквизиты
    ПО (ДоговорыКонтрагентов.Ссылка = ДоговорыКонтрагентовДополнительныеРеквизиты.Ссылка)
    И (ДоговорыКонтрагентовДополнительныеРеквизиты.Свойство.Имя = "ДатаСозданияДоговора_123")
    ГДЕ
    НЕ ДоговорыКонтрагентов.ПометкаУдаления
    И ДоговорыКонтрагентов.Статус = &Статус
    И ЕСТЬNULL(ДоговорыКонтрагентовДополнительныеРеквизиты.Значение, ДАТАВРЕМЯ(1,1,1)) = ДАТАВРЕМЯ(1,1,1)
```
---
СПАСИБО!!!